/*
	Estimates the degree of selection bias into the Glassdoor dataset for
	New Zealand college graduates employed in New Zealand
	Output:	Figure C1i
			Selection_NewZealand.csv
*/

local seed "C:\Users\jsock\Dropbox\Research\GD\International"

local dataPath "`seed'/Data"
local inputPath "`seed'/InputData"
local figurePath "`seed'/Replication/Figures"
local tablePath "`seed'/Replication/Tables"
local estimatePath "`seed'/Replication/Estimates"
local tempPath "`seed'/Replication/TempData"

********************************************************
* Save exchange rates as dta file
********************************************************

clear 
set more off
set matsize 10000

insheet using "`inputPath'/Exchange_rates_2022.csv", comma

keep country_glassdoor year ppp_xrat 

save `tempPath'/Exchange_rates.dta, replace 

local currentCountry = "New Zealand"  
local saveCountry = "NewZealand"

********************************************************
* Read in Australian undergraduate earnings
********************************************************

clear 
set more off
set matsize 10000

insheet using "`inputPath'/`saveCountry'_universities_earnings.csv", comma

foreach my_var of varlist median*{
	replace `my_var' = subinstr(`my_var',"$","",.)
	replace `my_var' = subinstr(`my_var',",","",.)
	destring `my_var' , replace
}

keep if median1 != .

generate country_glassdoor = "`currentCountry'"

merge m:1 country_glassdoor year using `tempPath'/Exchange_rates.dta

keep if _merge == 3
drop _merge

foreach my_var of varlist median*{
	replace `my_var' = `my_var' * ppp_xrat
}

sort school year

collapse (mean) median* , by(school)

save `tempPath'/`saveCountry'_universities_earnings_merge.dta, replace

********************************************************
* Read in country gdp to get universitycountry gdp
********************************************************

clear 
set more off
set matsize 10000
set scheme s1mono

* Set path and load data 
insheet using "`dataPath'/Salaries_international_dataset_main.csv", comma
drop v1

drop if jobtitle == ""

drop metro shortname 
drop city basecurrency country_iso dateval  
drop gender
drop sectorname iscurrentjobflag 

generate national_rank_pct = national_rank / numberuniversities

*--------------------------------
* Exclude users that leave more than 10 reviews
*--------------------------------

sort fk_userid yearofsalary 

by fk_userid : gen userReviews = _N

drop if userReviews > 10

drop userReviews

*--------------------------------
* Keep only Australia
*--------------------------------

scalar country_premia_thresh = 25

scalar selection_thresh = 25

scalar school_thresh = 25

drop if employertypecode == "SELF_EMPLOYED" 

keep if countryname == "`currentCountry'"

keep if universitycountry == "`currentCountry'"

*--------------------------------
* Convert with ppp exchange rates
*--------------------------------

generate logbase = ln(basesalary * ppp_xrat)

drop if logbase == .

*--------------------------------
* Exclude outliers in base pay
*--------------------------------

scalar scalingThresh = 10

generate realbase = basesalary * ppp_xrat
generate outside_thresh =  (realbase < (1/scalingThresh) * gdppw) | (realbase > scalingThresh * gdppw) 
drop realbase

drop if outside_thresh

*--------------------------------
* Generate indicators for valid in education analysis
*--------------------------------

generate hasDegree = degree != "UNMATCHED" & degree != "missing" & degree != ""  & degree != "HIGHSCHOOL"

generate uniDegree = degree == "BACHELORS" 

generate hasSchool = school != ""

generate valid_educ = uniDegree & hasDegree & hasSchool & universitycountry != ""

keep if valid_educ 

********************************************************
* Create valid sampels for one and two year comparisons
********************************************************

*--------------------------------
* Years since school ended
*--------------------------------

split endschool, p("-")

destring endschool1, gen(end_year)

generate years_from_graduate = yearofsalary - end_year

*--------------------------------
* Years since school ended
*--------------------------------

bys school : egen avg_grad_1 = median(logbase) if inrange(years_from_graduate,0,1)
bys school : egen avg_grad_3 = median(logbase) if inrange(years_from_graduate,2,3)
bys school : egen avg_grad_5 = median(logbase) if inrange(years_from_graduate,4,5)
bys school : egen avg_grad_7 = median(logbase) if inrange(years_from_graduate,6,7)
bys school : egen avg_grad_9 = median(logbase) if inrange(years_from_graduate,8,9)

bys school : gen n_grad_1 = sum(inrange(years_from_graduate,0,1))
bys school : gen n_grad_3 = sum(inrange(years_from_graduate,2,3))
bys school : gen n_grad_5 = sum(inrange(years_from_graduate,4,5))
bys school : gen n_grad_7 = sum(inrange(years_from_graduate,6,7))
bys school : gen n_grad_9 = sum(inrange(years_from_graduate,8,9))

*--------------------------------
* Add comparison earnings
*--------------------------------

merge m:1 school using `tempPath'/`saveCountry'_universities_earnings_merge.dta

*--------------------------------
* Comparet External data and GD data for recent graduates
*--------------------------------

preserve

	collapse (max) avg_grad* n_grad* median* national_rank national_rank_pct, by(school)
	
	* Stack earnings for different grad years 
	
	expand 5
	
	sort school

	generate gradYears = .
	by school: replace gradYears = 1 if _n == 1
	by school: replace gradYears = 3 if _n == 2
	by school: replace gradYears = 5 if _n == 3
	by school: replace gradYears = 7 if _n == 4
	by school: replace gradYears = 9 if _n == 5
	
	generate avg_grad = .
	replace avg_grad = avg_grad_1 if gradYears == 1
	replace avg_grad = avg_grad_3 if gradYears == 3
	replace avg_grad = avg_grad_5 if gradYears == 5
	replace avg_grad = avg_grad_7 if gradYears == 7
	replace avg_grad = avg_grad_9 if gradYears == 9
	
	generate undergrad_earnings = .
	replace undergrad_earnings = median1 if gradYears == 1
	replace undergrad_earnings = median3 if gradYears == 3
	replace undergrad_earnings = median5 if gradYears == 5
	replace undergrad_earnings = median7 if gradYears == 7
	replace undergrad_earnings = median9 if gradYears == 9
	
	generate n_grad = .
	replace n_grad = n_grad_1 if gradYears == 1
	replace n_grad = n_grad_3 if gradYears == 3
	replace n_grad = n_grad_5 if gradYears == 5
	replace n_grad = n_grad_7 if gradYears == 7
	replace n_grad = n_grad_9 if gradYears == 9
	
	generate avg_median_undergrad = ln(undergrad_earnings)
		
	* Difference between scorecard and gd 
	generate diff_grad = avg_grad - avg_median_undergrad	
	
	keep if diff_grad != .
	
	rename n_grad n_gd 
	bys school : egen n_grad = sum(n_gd)
	
	collapse (mean) diff_grad (max) n_grad national_rank_pct [aw=n_gd], by(school )
	
	* Get averages
	sum diff_grad
	local avg_equal = r(mean)
	scalar est_wtd_equal = r(mean)
	
	sum diff_grad [aw=n_grad]
	local avg_wtd_gd = r(mean)
	scalar est_wtd_gd = r(mean)
	scalar n_wt_wtd_gd = r(N)
	scalar sum_wt_wtd_gd = r(sum_w)

	* Get averages for top 5%
	sum diff_grad if national_rank_pct <= 0.05 [aw=n_grad]
	local avg_wtd_gd_top5 = r(mean)
	scalar est_wtd_gd_top5 = r(mean)
	scalar n_wt_wtd_gd_top5 = r(N)
	scalar sum_wt_wtd_gd_top5 = r(sum_w)
	
	* Get averages for not top 5%
	sum diff_grad if national_rank_pct > 0.05 | national_rank_pct == . [aw=n_grad]
	local avg_wtd_gd_not5 = r(mean)
	scalar est_wtd_gd_not5 = r(mean)
	scalar n_wt_wtd_gd_not5 = r(N)
	scalar sum_wt_wtd_gd_not5 = r(sum_w)

	* Weighted and unweighted kernel density
	twoway kdensity diff_grad [aw=n_grad], lcolor(gs3) lpattern(dash) xscale(r(-0.12(0.04)0.12)) xlabel(-0.12(0.04)0.12)  ///
		xline(0, lcolor(black) lpattern(solid) lwidth(thin)) fcolor(none) ///
		xtitle("log difference within college") ytitle("density") title("") scale(1.2)
	graph export "`figurePath'/Figure_C1i.eps"	, replace

restore

*--------------------------------
* Export summary of selection estimates
*--------------------------------

preserve

	generate selection_est_equal = est_wtd_equal
	generate selection_est_wtd = est_wtd_gd
	generate selection_est_wtd_top5 = est_wtd_gd_top5
	generate selection_est_wtd_not5 = est_wtd_gd_not5

	generate selection_wtd_N = n_wt_wtd_gd
	generate selection_wtd_N_top5 = n_wt_wtd_gd_top5
	generate selection_wtd_N_not5 = n_wt_wtd_gd_not5

	generate selection_wtd_sum = sum_wt_wtd_gd
	generate selection_wtd_sum_top5 = sum_wt_wtd_gd_top5
	generate selection_wtd_sum_not5 = sum_wt_wtd_gd_not5

	keep universitycountry selection_*
	
	keep if _n == 1
	
	order universitycountry selection_wtd_N* selection_wtd_sum* selection_est_* 
	
	outsheet using "`estimatePath'\Selection_`saveCountry'.csv" , comma replace

restore
